```
> restart:
  with(LinearAlgebra):
  with(plots):
  with(PDEtools):
  interface(rtablesize=20):
```

Linear elliptic PDEs: solution of Laplace and Poisson equations in 2D

The purpose of this worksheet is to illsutrate how to solve linear elliptic PDEs. For concreteness, we will focus on the following PDE:

$$\frac{\partial^2 u}{\partial x^2} + \frac{\partial^2 u}{\partial y^2} = f(x, y),$$

where the source function $f(x, y)$ is given. If $f(x, y) = 0$ this is known as Laplace's equation, if not it is Poisson's equation. We will seek a solution on a rectangular region of the xy-plane: $(x, y) \in [-L, L] \times [-L, L]$ subject to Dirichlet boundary conditions:

$$u(x, y=-L) = g_1(x), \quad u(x=+L, y) = g_2(y), \quad u(x, y=+L) = g_3(x), \quad u(x=-L, y) = g_4(y), .$$

Here, the g functions are assumed to be given. Here is a visualization of one choice of boundary conditions.

```
> p := 'p':
  L := 1:
  Range := -L..L;
  g[1] := x -> (1-x)*(1+x)^4;
  g[2] := y -> (y+L)/2/L;
  g[3] := x -> 2*Heaviside(x-1/2) - Heaviside(x+1/2);
  g[4] := y -> (y-L)*(y+L);

  p[1] := spacecurve([t,lhs(Range),g[1](t)],t=Range,axes=boxed,color=green):
  p[2] := spacecurve([rhs(Range),t,g[2](t)],t=Range,axes=boxed,color=red):
  p[3] := spacecurve([t,rhs(Range),g[3](t)],t=Range,axes=boxed,color=magenta):
  p[4] := spacecurve([lhs(Range),t,g[4](t)],t=Range,axes=boxed,color=blue):
  display(convert(p,list),labels=[x,y,u(x,y)],axes=framed,view=[-1.2*L..1.2*L,-1.2*L..1.2*L,
  default]);
```

$$Range := -1\,..1$$

$$g_1 := x \rightarrow (1-x)\,(1+x)^4$$

$$g_2 := y \to \frac{1}{2}\ \frac{y+L}{L}$$

$$g_3 := x \to 2\ \text{Heaviside}\left(x - \frac{1}{2}\right) - \text{Heaviside}\left(x + \frac{1}{2}\right)$$

$$g_4 := y \to (y - L)\ (y + L)$$



In this plot, the desired values of $u(x, y)$ are shown on the boundary of the region we are interested in. The goal of the numerical analysis will to be to "fill-in" the values of $u(x, y)$ interior to the boundary.

Numerical solution using a five-point stencil for the Laplacian

In this section, we will develop the solution to the following PDE assuming a stencil for the differential operator involving five points:

```
> pde := diff(u(x,y),x,x) + diff(u(x,y),y,y) - f(x,y);
```

$$pde := \frac{\partial^2}{\partial x^2} u(x, y) + \frac{\partial^2}{\partial y^2} u(x, y) - f(x, y) \quad (1.1$$

Stencil

As usual, we make use of the `GenerateStencil` procedure:

```
> GenerateStencil := proc(F,N,{orientation:=center,stepsize:=h,showorder:=true,showerror:=
  false})
        local vars, f, ii, Degree, stencil, Error, unknowns, Indets, ans, Phi, r, n, phi;

        Phi := convert(F,D);
        vars := op(Phi);
        n := PDEtools[difforder](Phi);
        f := op(1,op(0,Phi));
        if (nops([vars])<>1) then:
          r := op(1,op(0,op(0,Phi)));
        else:
          r := 1;
        fi:
        phi := f(vars);
        if (orientation=center) then:
          if (type(N,odd)) then:
            ii := [seq(i,i=-(N-1)/2..(N-1)/2)];
          else:
            ii := [seq(i,i=-(N-1)..(N-1),2)];
          fi;
        elif (orientation=left) then:
          ii := [seq(i,i=-N+1..0)]:
        elif (orientation=right) then:
          ii := [seq(i,i=0..N-1)]:
        fi;
        stencil := add(a[ii[i]]*subsop(r=op(r,phi)+ii[i]*stepsize,phi),i=1..N);
```

```
    Error := D[r$n](f)(vars) - stencil;
    Error := convert(series(Error,stepsize,N),polynom);
    unknowns := {seq(a[ii[i]],i=1..N)};
    Indets := indets(Error) minus {vars} minus unknowns minus {stepsize};
    Error := collect(Error,Indets,'distributed');
    ans := solve({coeffs(Error,Indets)},unknowns);
    if (ans=NULL) then:
       print(`Failure: try increasing the number of points in the stencil`);
       return NULL;
    fi:
    stencil := subs(ans,stencil);
    Error := convert(series(`leadterm`(D[r$n](f)(vars) - stencil),stepsize,N+20),
polynom);
    Degree := degree(Error,stepsize);
    if (showorder) then:
       print(cat(`This stencil is of order `,Degree));
    fi:
    if (showerror) then:
       print(cat(`This leading order term in the error is `,Error));
    fi:
    convert(D[r$n](f)(vars) = stencil,diff);

end proc:
```

We will use centered stencils for the second derivatives in the x and y directions:

```
> substencil[1] := GenerateStencil(diff(u(x,y),x,x),3);
  substencil[2] := GenerateStencil(diff(u(x,y),y,y),3);
```

This stencil is of order 2

$$substencil_1 := \frac{\partial^2}{\partial x^2} u(x,y) = \frac{u(x-h,y)}{h^2} - \frac{2\,u(x,y)}{h^2} + \frac{u(x+h,y)}{h^2}$$

This stencil is of order 2

$$substencil_2 := \frac{\partial^2}{\partial y^2} u(x,y) = \frac{u(x,y-h)}{h^2} - \frac{2\,u(x,y)}{h^2} + \frac{u(x,y+h)}{h^2} \qquad \textbf{(1.1.1)}$$

Notice that we have assumed the same stepsize h in the x and y directions. Putting these into **(1.1)** gives:

```
> stencil[1] := subs(substencil[1],substencil[2],pde);
```

$$stencil_1 := \frac{u(x-h,y)}{h^2} - \frac{4\,u(x,y)}{h^2} + \frac{u(x+h,y)}{h^2} + \frac{u(x,y-h)}{h^2} + \frac{u(x,y+h)}{h^2} - f(x,y) \qquad \textbf{(1.1.2)}$$

The first five terms are called the "five-point" stencil of the Laplacian operator in 2D since it involves evaluation of $u(x, y)$ at five different points. Here is a sketch of the relative orientation of these points.

```
> ngon := (n,x,y,r,phi) -> [seq([x+r*cos(2*Pi*i/n+phi), y+r*sin(2*Pi*i/n+phi)], i = 1 ..
  n)]:
  display([seq(polygonplot(ngon(20,i,0,0.1,Pi/2),color=red),i=-1..1),polygonplot(ngon(20,
  0,1,0.1,Pi/8),color=red),polygonplot(ngon(20,0,-1,0.1,Pi/8),color=red),textplot([seq([i,
  -0.1,typeset(u(x+i*h,y))],i=-1..1),[0,0.9,typeset(u(x,y+h))],[0,-1.1,typeset(u(x,y-h))]
  ],align={below})],view=[-1.4..1.4,-1.4..1.4],tickmarks=[[-1=x-h,0=x,1=x+h],[-1=y-h,0=y,
  1=y+h]],axes=boxed,scaling=constrained,labels=["",""],title="Five point stencil of the
  Laplacian");
```



Let's check the error in the stencil by expanding in a Taylor series and then making use of the original PDE:

```
> Error := series(stencil[1],h,8):
  Error := convert(dsubs(isolate(pde,f(x,y)),Error),D);
```

$$Error := \left(\frac{1}{12} D_{1,1,1,1}(u)(x,y) + \frac{1}{12} D_{2,2,2,2}(u)(x,y)\right) h^2 + \left(\frac{1}{360} D_{2,2,2,2,2,2}(u)(x,y) \qquad \textbf{(1.1.3}$$

$$+\frac{1}{360}D_{1,1,1,1,1,1}(u)(x,y)\Big)h^4+O(h^6)$$

It is common in the literature to conclude from this result that the error in the stencil is $O(h^2)$. However, some care is warranted with this terminology since it relies on the stencil being written precisely in the form of **(1.1.2)**; i.e., on the u terms all being divided by h^2. If we had instead written our stencil as:

```
> alt_stencil := expand(h^2*stencil[1]);
```

$$alt\_stencil := u(x-h,y)-4u(x,y)+u(x+h,y)+u(x,y-h)+u(x,y+h)-h^2 f(x,y) \quad \textbf{(1.1.4}$$

and expanded about $h=0$ the leading order behaviour would have been $O(h^4)$:

```
> alt_Error := series(alt_stencil,h,6):
  alt_Error := convert(dsubs(isolate(pde,f(x,y)),alt_Error),D);
```

$$alt\_Error := \left(\frac{1}{12}D_{1,1,1,1}(u)(x,y)+\frac{1}{12}D_{2,2,2,2}(u)(x,y)\right)h^4+O(h^6) \quad \textbf{(1.1.5}$$

The common definition of error for the numeric solution PDEs involves first writing the stencil in a form whose limit is the original PDE as the cell size goes to zero:

```
> limit(stencil[1],h=0);
  limit(alt_stencil,h=0);
```

$$D_{2,2}(u)(x,y)+D_{1,1}(u)(x,y)-f(x,y)$$

$$0 \quad \textbf{(1.1.6}$$

We see that `stencil` matches this requirement, but `alt_stencil` does not. The error is then defined as the discrepancy between the stencil written in the perferred form and zero; i.e., what we have called `Error` but not what we've called `alt_Error`.

The algorithm

Having now obtained a discrete form **(1.1.2)** of the PDE **(1.1)**, we now turn our attention to how to exploit it and obtain a numeric solution. We first need to discretize the domain $(x,y)\in[a,b]\times[c,d]$ over which we seek a solution. We will assume $N+2$ lattice points in the x and y directions, respectively. The coordinates of the lattice point will be explicitly given by $x_i = Z(i)$ and $y_j = Z(j)$, where

```
> L := 'L':
  N := 'N':
  g := 'g':
  Z := i -> -L+2*L/(N+1)*i;
  x[0] = Z(0),x[N+1] = Z(N+1),y[0] = Z(0),y[N+1] = Z(N+1);
```

$$Z := i \to -L + \frac{2\,L\,i}{N+1}$$

$$x_0 = -L, x_{N+1} = L, y_0 = -L, y_{N+1} = L \qquad \textbf{(1.2.1}$$

Here is a visualization of the lattice:

```
> N := 4;
  L := 1;
  r := L/(N+1)/4;
  ngon := (n,x,y,r,phi) -> [seq([x+r*cos(2*Pi*i/n+phi), y+r*sin(2*Pi*i/n+phi)], i = 1 ..
  n)]:
  p[1] := display([seq(polygonplot(ngon(4,Z(0),Z(j),r,Pi/2),color=magenta),j=0..N+1),seq
  (polygonplot(ngon(4,Z(N+1),Z(j),r,Pi/2),color=magenta),j=0..N+1),seq(polygonplot(ngon(4,
  Z(i),Z(0),r,0),color=magenta),i=1..N),seq(polygonplot(ngon(4,Z(i),Z(N+1),r,0),color=
  magenta),i=1..N),seq(seq(polygonplot(ngon(20,Z(i),Z(j),r,0),color=white),i=1..N),j=1..
  N),textplot([seq(seq([Z(i+0.1),Z(j),typeset(u[i,j])],i=0..N+1),j=0..N+1)],align={above,
  right})],view=[Z(-1)..Z(N+2),Z(-1)..Z(N+2)],tickmarks=[[seq(Z(i)=typeset(x[i]=evalf[2](Z
  (i))),i=0..N+1)],[seq(Z(i)=typeset(y[i]=evalf[2](Z(i))),i=0..N+1)]],axes=boxed,scaling=
  constrained,labels=[``,``]):
  p[1];
```

$$N := 4$$

$$L := 1$$

$$r := \frac{1}{20}$$



$u_{0,5}$ $u_{1,5}$ $u_{2,5}$ $u_{3,5}$ $u_{4,5}$ $u_{5,5}$
$u_{0,4}$ $u_{1,4}$ $u_{2,4}$ $u_{3,4}$ $u_{4,4}$ $u_{5,4}$
$u_{0,3}$ $u_{1,3}$ $u_{2,3}$ $u_{3,3}$ $u_{4,3}$ $u_{5,3}$
$u_{0,2}$ $u_{1,2}$ $u_{2,2}$ $u_{3,2}$ $u_{4,2}$ $u_{5,2}$
$u_{0,1}$ $u_{1,1}$ $u_{2,1}$ $u_{3,1}$ $u_{4,1}$ $u_{5,1}$
$u_{0,0}$ $u_{1,0}$ $u_{2,0}$ $u_{3,0}$ $u_{4,0}$ $u_{5,0}$
$y_5 = 1.$
$y_4 = 0.6$
$y_3 = 0.2$
$y_2 = -0.2$
$y_1 = -0.6$
$y_0 = -1.$
$x_0 = -1.$
$x_1 = -0.6$
$x_2 = -0.2$
$x_3 = 0.2$
$x_4 = 0.6$
$x_5 = 1.$

Each node is labelled by our approximation to the true solution of the PDE $u_{i,j} \approx u(x_i, y_j)$. Boundary conditions will be used to fix $u_{i,j}$ at each of the purple boundary nodes, so the goal of the code will be to solve for $u_{i,j}$ at each of the white interior nodes. We also define $f_{i,j} = f(x_i, y_j)$, which allows us to re-write the stencil **(1.1.2)** as:

```
> Subs := seq(seq(u(x+ii*h,y+jj*h)=u[i+ii,j+jj],ii=-1..1),jj=-1..1),f(x,y)=f[i,j]:
  stencil[1] := subs(Subs,stencil[1]);
```

$$stencil_1 := \frac{u_{i-1,j}}{h^2} - \frac{4\,u_{i,j}}{h^2} + \frac{u_{i+1,j}}{h^2} + \frac{u_{i,j-1}}{h^2} + \frac{u_{i,j+1}}{h^2} - f_{i,j} \quad \textbf{(1.2.2)}$$

We will enforce this stencil for $i = 1 \ldots N$ and $j = 1 \ldots N$; i.e., at each of the white circles in the above plot. To do this, it is useful to rewrite the stencil as a mapping:

```
> Stencil[1] := unapply(stencil[1],h,i,j,u,f);
```

$$Stencil_1 := (h, i, j, u, f) \rightarrow \frac{u_{i-1,j}}{h^2} - \frac{4\,u_{i,j}}{h^2} + \frac{u_{i+1,j}}{h^2} + \frac{u_{i,j-1}}{h^2} + \frac{u_{i,j+1}}{h^2} - f_{i,j} \quad \textbf{(1.2.3)}$$

Here is an example of the kind of system of equations we need to solve for $N^2 = 9$ total internal lattice points. The boundary conditions $u(x, -L) = g_1(x)$, etc. are implemented by assigning values to all the boundary nodes (purple diamonds in the above sketch). For example, we set $u_{0,i} = g_1(x_i)$. The boundary conditions are held in the list **BCs**.

Aside: There is a potential conflict at the corners of our lattice where $(i, j) = (0, 0), (N+1, 0), (N+1, N+1), (0, N+1)$ if $g_1(L) \neq g_2(-L), g_2(-L) \neq g_3(L), g_3(-L) \neq g_4(L)$, or $g_4(-L) \neq g_1(-L)$. Ideally, we should choose boundary data to ensure that this does not happen, but if there is an ambiguity we will (arbitrarily) assume that the top and bottom boundary data (g_1 and g_3) take precedence over the left and right boundary data (g_2 and g_4).

```
> N := 3:

  BCs := [seq(u[0,i]  =g[4](y[i]),i=1..N),
          seq(u[i,0]  =g[1](x[i]),i=0..N+1),
          seq(u[N+1,i]=g[2](y[i]),i=1..N),
          seq(u[i,N+1]=g[3](x[i]),i=0..N+1)]:

  sys := Vector([subs(BCs,[seq(seq(Stencil[1](h,i,j,u,f),i=1..N),j=1..N)])]);
```

$$sys := \left[\begin{array}{c} \frac{g_4(y_1)}{h^2} - \frac{4\,u_{1,1}}{h^2} + \frac{u_{2,1}}{h^2} + \frac{g_1(x_1)}{h^2} + \frac{u_{1,2}}{h^2} - f_{1,1} \\ \frac{u_{1,1}}{h^2} - \frac{4\,u_{2,1}}{h^2} + \frac{u_{3,1}}{h^2} + \frac{g_1(x_2)}{h^2} + \frac{u_{2,2}}{h^2} - f_{2,1} \\ \frac{u_{2,1}}{h^2} - \frac{4\,u_{3,1}}{h^2} + \frac{g_2(y_1)}{h^2} + \frac{g_1(x_3)}{h^2} + \frac{u_{3,2}}{h^2} - f_{3,1} \\ \frac{g_4(y_2)}{h^2} - \frac{4\,u_{1,2}}{h^2} + \frac{u_{2,2}}{h^2} + \frac{u_{1,1}}{h^2} + \frac{u_{1,3}}{h^2} - f_{1,2} \\ \frac{u_{1,2}}{h^2} - \frac{4\,u_{2,2}}{h^2} + \frac{u_{3,2}}{h^2} + \frac{u_{2,1}}{h^2} + \frac{u_{2,3}}{h^2} - f_{2,2} \\ \frac{u_{2,2}}{h^2} - \frac{4\,u_{3,2}}{h^2} + \frac{g_2(y_2)}{h^2} + \frac{u_{3,1}}{h^2} + \frac{u_{3,3}}{h^2} - f_{3,2} \\ \frac{g_4(y_3)}{h^2} - \frac{4\,u_{1,3}}{h^2} + \frac{u_{2,3}}{h^2} + \frac{u_{1,2}}{h^2} + \frac{g_3(x_1)}{h^2} - f_{1,3} \\ \frac{u_{1,3}}{h^2} - \frac{4\,u_{2,3}}{h^2} + \frac{u_{3,3}}{h^2} + \frac{u_{2,2}}{h^2} + \frac{g_3(x_2)}{h^2} - f_{2,3} \\ \frac{u_{2,3}}{h^2} - \frac{4\,u_{3,3}}{h^2} + \frac{g_2(y_3)}{h^2} + \frac{u_{3,2}}{h^2} + \frac{g_3(x_3)}{h^2} - f_{3,3} \end{array}\right] \quad (1.2.4$$

This is a linear system in the $u_{i,j}$'s. Though it is natural to think of the $u_{i,j}$'s to be arranged in a matrix, it is more convenient to reshape them into a vector as indicated by this before and after plot:

```
> p[1] := display([seq(polygonplot(ngon(4,Z(0),Z(j),r,Pi/2),color=magenta),j=0..N+1),seq
  (polygonplot(ngon(4,Z(N+1),Z(j),r,Pi/2),color=magenta),j=0..N+1),seq(polygonplot(ngon(4,
  Z(i),Z(0),r,0),color=magenta),i=1..N),seq(polygonplot(ngon(4,Z(i),Z(N+1),r,0),color=
  magenta),i=1..N),seq(seq(polygonplot(ngon(20,Z(i),Z(j),r,0),color=white),i=1..N),j=1..
  N),textplot([seq(seq([Z(i+0.1),Z(j),typeset(u[i,j])],i=1..N),j=1..N)],align={above,
  right})],view=[Z(-1)..Z(N+2),Z(-1)..Z(N+2)],tickmarks=[[seq(Z(i)=typeset(x[i]=evalf[2](Z
```

```
(i))),i=0..N+1)],[seq(Z(i)=typeset(y[i]=evalf[2](Z(i))),i=0..N+1)]],axes=boxed,scaling=
constrained,labels=[``,``],title="Before re-labelling"):

p[2] := display([seq(polygonplot(ngon(4,Z(0),Z(j),r,Pi/2),color=magenta),j=0..N+1),seq
(polygonplot(ngon(4,Z(N+1),Z(j),r,Pi/2),color=magenta),j=0..N+1),seq(polygonplot(ngon(4,
Z(i),Z(0),r,0),color=magenta),i=1..N),seq(polygonplot(ngon(4,Z(i),Z(N+1),r,0),color=
magenta),i=1..N),seq(seq(polygonplot(ngon(20,Z(i),Z(j),r,0),color=white),i=1..N),j=1..
N),textplot([seq(seq([Z(i+0.1),Z(j),typeset(w[(j-1)*N+i])],i=1..N),j=1..N)],align=
{above,right})],view=[Z(-1)..Z(N+2),Z(-1)..Z(N+2)],tickmarks=[[seq(Z(i)=typeset(x[i]=
evalf[2](Z(i))),i=0..N+1)],[seq(Z(i)=typeset(y[i]=evalf[2](Z(i))),i=0..N+1)]],axes=
boxed,scaling=constrained,labels=[``,``],title="After re-labelling"):

display(Array(1..2,[p[1],p[2]]));
```



In other words, we re-label the $u_{i,j}$ sequentially by rows. With this understanding **(1.2.4)** is seen to be a linear system **Aw = b** with

```
> w := [seq(seq(u[i,j],i=1..N),j=1..N)];
  A,b := GenerateMatrix(convert(sys,list),w);
```

$$w := [u_{1,1}, u_{2,1}, u_{3,1}, u_{1,2}, u_{2,2}, u_{3,2}, u_{1,3}, u_{2,3}, u_{3,3}]$$

$$A, b := \begin{bmatrix} -\frac{4}{h^2} & \frac{1}{h^2} & 0 & \frac{1}{h^2} & 0 & 0 & 0 & 0 & 0 \\ \frac{1}{h^2} & -\frac{4}{h^2} & \frac{1}{h^2} & 0 & \frac{1}{h^2} & 0 & 0 & 0 & 0 \\ 0 & \frac{1}{h^2} & -\frac{4}{h^2} & 0 & 0 & \frac{1}{h^2} & 0 & 0 & 0 \\ \frac{1}{h^2} & 0 & 0 & -\frac{4}{h^2} & \frac{1}{h^2} & 0 & \frac{1}{h^2} & 0 & 0 \\ 0 & \frac{1}{h^2} & 0 & \frac{1}{h^2} & -\frac{4}{h^2} & \frac{1}{h^2} & 0 & \frac{1}{h^2} & 0 \\ 0 & 0 & \frac{1}{h^2} & 0 & \frac{1}{h^2} & -\frac{4}{h^2} & 0 & 0 & \frac{1}{h^2} \\ 0 & 0 & 0 & \frac{1}{h^2} & 0 & 0 & -\frac{4}{h^2} & \frac{1}{h^2} & 0 \\ 0 & 0 & 0 & 0 & \frac{1}{h^2} & 0 & \frac{1}{h^2} & -\frac{4}{h^2} & \frac{1}{h^2} \\ 0 & 0 & 0 & 0 & 0 & \frac{1}{h^2} & 0 & \frac{1}{h^2} & -\frac{4}{h^2} \end{bmatrix}, \begin{bmatrix} -\frac{g_4(y_1)}{h^2} + f_{1,1} - \frac{g_1(x_1)}{h^2} \\ f_{2,1} - \frac{g_1(x_2)}{h^2} \\ f_{3,1} - \frac{g_2(y_1)}{h^2} - \frac{g_1(x_3)}{h^2} \\ -\frac{g_4(y_2)}{h^2} + f_{1,2} \\ f_{2,2} \\ f_{3,2} - \frac{g_2(y_2)}{h^2} \\ -\frac{g_4(y_3)}{h^2} + f_{1,3} - \frac{g_3(x_1)}{h^2} \\ -\frac{g_3(x_2)}{h^2} + f_{2,3} \\ -\frac{g_3(x_3)}{h^2} + f_{3,3} - \frac{g_2(y_3)}{h^2} \end{bmatrix} \quad (1.2.5$$

To obtain our numeric solution for $u(x, y)$ all we need to do is solve this system for **w** and hence $u_{i,j}$. Here is some code that performs this task (we set the **choice** parameter equal to 1 to use the five-point stencil discussed here; later we set it equal to 2 to use the nine-point stencil):

```
> PoissonSolve := proc(N,_f,g,L,choice)
    local Z, h, i, u, f, sys, w, sol, j, Data:

    # define basic grid parameters
    Z := i -> -L+2*L/(N+1)*i;
    h := evalf(Z(1)-Z(0));
```

```
    # fix the boundary data and the source matrix
    for i from 0 to N+1 do:
      u[N+1,i] := evalf(g[2](Z(i)));
      u[0,i] := evalf(g[4](Z(i)));
      u[i,0] := evalf(g[1](Z(i)));
      u[i,N+1] := evalf(g[3](Z(i)));
    od:
    f := Array(0..N+1,0..N+1,[seq([seq(evalf(_f(Z(i),Z(j))),i=0..N+1)],j=0..N+1)],
 datatype=float);

    # write down the system of equations to solve and solve them
    sys := [seq(seq(Stencil[choice](h,i,j,u,f),i=1..N),j=1..N)];
    w := [seq(seq(u[i,j],i=1..N),j=1..N)];
    sol := LinearSolve(GenerateMatrix(sys,w));

    # parse the solution vector sol back into "matrix" form
    for i from 1 to N do:
      for j from 1 to N do:
         u[i,j] := sol[(j-1)*N+i]:
      od:
    od:

    # generate a 3D plot of the solution using the surfdata command
    Data := [seq([seq([Z(i),Z(j),u[i,j]],i=0..N+1)],j=0..N+1)]:
    surfdata(Data,axes=boxed,labels=[`x`,`y`,`u(x,y)`],shading=zhue,style=patchcontour);

 end proc:
```

Here is an example of the output when the source function is set to zero $f(x, y) = 0$; i.e., when **(1.1)** reduces down to Laplace's equation

```
> f := (x,y) -> 0;
  g := [x -> 0,x -> 0,x -> 0,x -> (1-x)*(1+x)];
  PoissonSolve(10,f,g,1,1);
```

$$f := (x, y) \to 0$$

$$g := [x \to 0, x \to 0, x \to 0, x \to (1 - x)\,(1 + x)]$$



Here is an example problem the calculates the electric potential around a ring of $2n$ alternating charges. We model the charges as discs of radius r located a distance of R away from the origin and with surface charge density ±1. (For $n = 1$, this is an electric dipole.)

```
> n := 2;
  r := 4;
  R := 7;
  L := 25;
  g := [x->0,x->0,x->0,x->0];
```

```
f := (x,y) -> add((-1)^i*Heaviside(r^2-(x-R*cos(Pi*i/n))^2 - (y-R*sin(Pi*i/n))^2),i=0.
.2*n-1);
p[1] := plot3d(f(x,y),x=-L..L,y=-L..L,grid=[100,100],shading=zhue,axes=boxed,style=
patchcontour,labels=[x,y,`f(x,y)`]):
p[2] := PoissonSolve(30,f,g,L,1):
display(Array([p[1],p[2]]));
```

$$n := 2$$

$$r := 4$$

$$R := 7$$

$$L := 25$$

$$g := [x \to 0, x \to 0, x \to 0, x \to 0]$$

$$f := (x, y) \to add\left((-1)^i \,\mathrm{Heaviside}\left(r^2 - \left(x - R\cos\left(\frac{\pi\, i}{n}\right)\right)^2 - \left(y - R\sin\left(\frac{\pi\, i}{n}\right)\right)^2\right), i = 0 ..2\, n - 1\right)$$



Numerical solution using a nine-point stencil for the Laplacian

```
> f := 'f';
  Error := 'Error':
  sys := 'sys':
```

$$f := f \quad (2.1$$

We now attempt to find a more accurate stencil to the PDE **(1.1)**. Let us assume a nine point stencil for the Laplacian of the form:

```
> substencil[3] := diff(u(x,y),x,x)+diff(u(x,y),y,y) = add(add(a[i,j]/h^2*u(x+i*h,y+j*h),i=
  -1..1),j=-1..1);
  stencil[2] := subs(isolate(substencil[3],diff(u(x,y),x,x)),pde);
```

$$substencil_3 := \frac{\partial^2}{\partial x^2} u(x,y) + \frac{\partial^2}{\partial y^2} u(x,y) = \frac{a_{-1,-1}\, u(x-h, y-h)}{h^2} + \frac{a_{0,-1}\, u(x, y-h)}{h^2} + \frac{a_{1,-1}\, u(x+h, y-h)}{h^2} + \frac{a_{-1,0}\, u(x-h, y)}{h^2} + \frac{a_{0,0}\, u(x,y)}{h^2} + \frac{a_{1,0}\, u(x+h, y)}{h^2} + \frac{a_{-1,1}\, u(x-h, y+h)}{h^2} + \frac{a_{0,1}\, u(x, y+h)}{h^2} + \frac{a_{1,1}\, u(x+h, y+h)}{h^2}$$

$$stencil_2 := \frac{a_{-1,-1}\, u(x-h, y-h)}{h^2} + \frac{a_{0,-1}\, u(x, y-h)}{h^2} + \frac{a_{1,-1}\, u(x+h, y-h)}{h^2} + \frac{a_{-1,0}\, u(x-h, y)}{h^2} + \frac{a_{0,0}\, u(x,y)}{h^2} + \frac{a_{1,0}\, u(x+h, y)}{h^2} + \frac{a_{-1,1}\, u(x-h, y+h)}{h^2} + \frac{a_{0,1}\, u(x, y+h)}{h^2} + \frac{a_{1,1}\, u(x+h, y+h)}{h^2} - f(x,y) \qquad \textbf{(2.2}$$

This stencil involves a square array of points centered about (x, y), as shown in the plot:

```
> display([seq(seq(polygonplot(ngon(20,i,j,0.1,Pi/2),color=red),i=-1..1),j=-1..1),textplot(
  [seq([i,-0.1,typeset(u(x+i*h,y))],i=-1..1),seq([i,0.9,typeset(u(x,y+h))],i=-1..1),seq([i,
  -1.1,typeset(u(x,y-h))],i=-1..1)],align={below})],view=[-1.4..1.4,-1.4..1.4],tickmarks=[
  [-1=x-h,0=x,1=x+h],[-1=y-h,0=y,1=y+h]],axes=boxed,scaling=constrained,labels=["",""],
  title="Nine point stencil of the Laplacian");
```



We will now calculate the error in the stencil, simplify the expression using the PDE **(1.1)**, and then select the $a_{i,j}$ coefficients to minimize the error. More specifically, we'll try to eliminate the derivatives of $u(x, y)$ of order 2 and higher from the Taylor series of **(2.2)** using the following relations:

```
> derivative[1] := isolate(pde,diff(u(x,y),x,x));
  derivative[2] := diff(derivative[1],x);
  derivative[3] := diff(derivative[2],x);
```

$$derivative_1 := \frac{\partial^2}{\partial x^2} u(x,y) = -\left(\frac{\partial^2}{\partial y^2} u(x,y)\right) + f(x,y)$$

$$derivative_2 := \frac{\partial^3}{\partial x^3} u(x,y) = -\left(\frac{\partial^3}{\partial y^2 \partial x} u(x,y)\right) + \frac{\partial}{\partial x} f(x,y)$$

$$derivative_3 := \frac{\partial^4}{\partial x^4} u(x,y) = -\left(\frac{\partial^4}{\partial y^2 \partial x^2} u(x,y)\right) + \frac{\partial^2}{\partial x^2} f(x,y) \quad \textbf{(2.3}$$

Here is our calculation of the error in the stencil **(2.2)**. The set `sys[Poisson]` is the set of equations to be satisfied by the $a_{i,j}$'s to ensure that the error is $O(h^3)$ or higher.

```
> Error[Poisson] := convert(convert(series(stencil[2],h,5),polynom),diff):
  Error[Poisson] := dsubs(convert(derivative,list),Error[Poisson]):
  vars := {seq(seq(a[i,j],i=-1..1),j=-1..1)}:
  Indets := indets(Error[Poisson]) minus vars minus {x,y};
  Error[Poisson] := collect(Error[Poisson],Indets,'distributed'):
  sys[Poisson] := {coeffs(Error[Poisson],Indets)};
```

$$Indets := \left\{h, \frac{\partial}{\partial x} f(x,y), \frac{\partial}{\partial y} f(x,y), \frac{\partial^2}{\partial x^2} f(x,y), \frac{\partial^2}{\partial y^2} f(x,y), \frac{\partial}{\partial x} u(x,y), \frac{\partial}{\partial y} u(x,y), \frac{\partial^2}{\partial y^2} u(x,y), \frac{\partial^3}{\partial y^3} u(x,y), \frac{\partial^4}{\partial y^4} u(x,y), \frac{\partial^2}{\partial y \partial x} f(x,y), \frac{\partial^2}{\partial y \partial x} u(x,y), \frac{\partial^3}{\partial y^2 \partial x} u(x,y), f(x,y), u(x,y)\right\}$$

$$sys_{Poisson} := \Big\{a_{-1,-1} - a_{1,-1} - a_{-1,1} + a_{1,1}, -\frac{1}{6} a_{-1,1} + \frac{1}{6} a_{-1,-1} - \frac{1}{6} a_{1,-1} + \frac{1}{6} a_{1,1}, \frac{1}{2} a_{-1,1} - \frac{1}{2} a_{-1,-1} - \frac{1}{2} a_{1,-1} + \frac{1}{2} a_{1,1}, -\frac{1}{2} a_{1,0} - \frac{1}{2} a_{-1,0} + \frac{1}{2} a_{0,1} + \frac{1}{2} a_{0,-1}, -a_{-1,-1} + a_{1,0} + a_{1,1} + a_{1,-1} - a_{-1,0} - a_{-1,1}, -\frac{1}{3} a_{-1,-1} - \frac{1}{6} a_{1,0} + \frac{1}{3} a_{1,1} + \frac{1}{6} a_{-1,0} - \frac{1}{3} a_{-1,1} + \frac{1}{3} a_{1,-1}, -\frac{1}{6} a_{-1,0} + \frac{1}{6} a_{1,0} + \frac{1}{6} a_{1,1} - \frac{1}{6} a_{-1,1} - \frac{1}{6} a_{-1,-1} + \frac{1}{6} a_{1,-1}, -\frac{1}{24} a_{-1,0} - \frac{1}{24} a_{1,0} + \frac{5}{24} a_{1,-1} + \frac{5}{24} a_{-1,1} + \frac{5}{24} a_{-1,-1} + \frac{5}{24} a_{1,1}, -a_{0,-1} - a_{-1,-1} - a_{1,-1} + a_{0,1} + a_{1,1} + a_{-1,1}, -\frac{1}{6} a_{0,-1} - \frac{1}{3} a_{-1,1} - \frac{1}{3} a_{1,1} + \frac{1}{3} a_{1,-1} + \frac{1}{6} a_{0,1} + \frac{1}{3} a_{-1,-1}, \frac{1}{24} a_{1,0} + \frac{1}{24} a_{1,-1} + \frac{1}{24} a_{-1,0} + \frac{1}{24} a_{-1,1} + \frac{1}{24} a_{-1,-1} + \frac{1}{24} a_{1,1}, \frac{1}{2} a_{-1,0} - 1 + \frac{1}{2} a_{-1,1} + \frac{1}{2} a_{-1,-1} + \frac{1}{2} a_{1,-1} \quad \textbf{(2.4}$$

$$+\frac{1}{2}a_{1,0}+\frac{1}{2}a_{1,1}, \frac{1}{24}a_{1,0}+\frac{1}{24}a_{-1,0}+\frac{1}{24}a_{0,-1}-\frac{1}{6}a_{1,-1}-\frac{1}{6}a_{1,1}+\frac{1}{24}a_{0,1}-\frac{1}{6}a_{-1,1}-\frac{1}{6}a_{-1,-1},$$
$$a_{-1,-1}+a_{-1,0}+a_{0,-1}+a_{1,1}+a_{1,-1}+a_{1,0}+a_{-1,1}+a_{0,0}+a_{0,1}\Big\}$$

This a linear system of 14 equations for 9 unknowns; there is no solution.

```
> nops(sys[Poisson]);
  nops(vars);
  solve(sys[Poisson]);
```

$$14$$
$$9 \qquad (2.5$$

So the stencil **(2.2)** cannot be made to yield an error smaller that $O(h^2)$ when solving Poisson's equation. This is no better than the five point stencil of the previous section. However if we now specialize to Laplace's equation by setting $f(x,y)=0$, we can do a little better:

```
> Error[laplace] := dsubs(f(x,y)=0,Error[Poisson]):
  vars := {seq(seq(a[i,j],i=-1..1),j=-1..1)}:
  Indets := indets(Error[laplace]) minus vars minus {x,y};
  Error[laplace] := collect(Error[laplace],Indets,'distributed'):
  sys[laplace] := {coeffs(Error[laplace],Indets)};
```

$$Indets := \left\{h, \frac{\partial}{\partial x}u(x,y), \frac{\partial}{\partial y}u(x,y), \frac{\partial^2}{\partial y^2}u(x,y), \frac{\partial^3}{\partial y^3}u(x,y), \frac{\partial^4}{\partial y^4}u(x,y), \frac{\partial^2}{\partial y\,\partial x}u(x,y), \frac{\partial^3}{\partial y^2\,\partial x}u(x,y), u(x,y)\right\}$$

$$sys_{laplace} := \Big\{a_{-1,-1}-a_{1,-1}-a_{-1,1}+a_{1,1}, -\frac{1}{2}a_{1,0}-\frac{1}{2}a_{-1,0}+\frac{1}{2}a_{0,1}+\frac{1}{2}a_{0,-1}, -a_{-1,-1}+a_{1,0}+a_{1,1}+a_{1,-1} \qquad (2.6$$
$$-a_{-1,0}-a_{-1,1}, -\frac{1}{3}a_{-1,-1}-\frac{1}{6}a_{1,0}+\frac{1}{3}a_{1,1}+\frac{1}{6}a_{-1,0}-\frac{1}{3}a_{-1,1}+\frac{1}{3}a_{1,-1}, -a_{0,-1}-a_{-1,-1}-a_{1,-1}+a_{0,1}$$
$$+a_{1,1}+a_{-1,1}, -\frac{1}{6}a_{0,-1}-\frac{1}{3}a_{-1,1}-\frac{1}{3}a_{1,1}+\frac{1}{3}a_{1,-1}+\frac{1}{6}a_{0,1}+\frac{1}{3}a_{-1,-1}, \frac{1}{24}a_{1,0}+\frac{1}{24}a_{-1,0}$$
$$+\frac{1}{24}a_{0,-1}-\frac{1}{6}a_{1,-1}-\frac{1}{6}a_{1,1}+\frac{1}{24}a_{0,1}-\frac{1}{6}a_{-1,1}-\frac{1}{6}a_{-1,-1}, a_{-1,-1}+a_{-1,0}+a_{0,-1}+a_{1,1}+a_{1,-1}+a_{1,0}$$
$$+a_{-1,1}+a_{0,0}+a_{0,1}\Big\}$$

We now have 8 equations for 9 unknowns, yielding a one parameter family of solutions for the coefficients:

```
> nops(sys[laplace]);
  nops(vars);
```

```
solve(sys[laplace]);
```

$$\frac{8}{9}$$

$$\{a_{-1,-1}=a_{1,1}, a_{-1,0}=4\,a_{1,1}, a_{-1,1}=a_{1,1}, a_{0,-1}=4\,a_{1,1}, a_{0,0}=-20\,a_{1,1}, a_{0,1}=4\,a_{1,1}, a_{1,-1}=a_{1,1}, a_{1,0}=4\,a_{1,1}, a_{1,1}=a_{1,1}\} \tag{2.7}$$

What is the reason behind the different sizes of `sys[Poisson]` and `sys[laplace]`? Basically, there are more indeterminants in the Poisson equation error from the source function (i.e., $f(x,y), f_x(x,y), f_y(x,y)$, etc.) resulting in more equations that must be satisfied to cancel all the terms of order h^2 or less. The classic nine-point stencil is defined by the following assumption (of course, this is arbitrary and can be changed):

```
> assumptions := {a[1,1]=1/6};
  ans := solve(subs(assumptions,sys[laplace])) union assumptions;
```

$$assumptions := \left\{a_{1,1}=\frac{1}{6}\right\}$$

$$ans := \left\{a_{-1,-1}=\frac{1}{6}, a_{-1,0}=\frac{2}{3}, a_{-1,1}=\frac{1}{6}, a_{0,-1}=\frac{2}{3}, a_{0,0}=-\frac{10}{3}, a_{0,1}=\frac{2}{3}, a_{1,-1}=\frac{1}{6}, a_{1,0}=\frac{2}{3}, a_{1,1}=\frac{1}{6}\right\} \tag{2.8}$$

We sub this back into **(2.2)**:

```
> stencil[2] := subs(ans,stencil[2]);
```

$$stencil_2 := \frac{1}{6}\,\frac{u(x-h,y-h)}{h^2}+\frac{2}{3}\,\frac{u(x,y-h)}{h^2}+\frac{1}{6}\,\frac{u(x+h,y-h)}{h^2}+\frac{2}{3}\,\frac{u(x-h,y)}{h^2}-\frac{10}{3}\,\frac{u(x,y)}{h^2}+\frac{2}{3}\,\frac{u(x+h,y)}{h^2}+\frac{1}{6}\,\frac{u(x-h,y+h)}{h^2}+\frac{2}{3}\,\frac{u(x,y+h)}{h^2}+\frac{1}{6}\,\frac{u(x+h,y+h)}{h^2}-f(x,y) \tag{2.9}$$

Note that we have left the source term in this expression; the reason will be apparent shortly. Let's calculate the error with these specific values of the coefficients:

```
> Error[poisson] := dsubs(convert(derivative,list),convert(convert(series(stencil[2],h,6),`+
  `),diff));
```

$$Error_{poisson} := \frac{1}{12}\,h^2\left(\frac{\partial^2}{\partial y^2}f(x,y)\right)+\frac{1}{12}\,h^2\left(\frac{\partial^2}{\partial x^2}f(x,y)\right)+\mathrm{O}(h^4) \tag{2.10}$$

We immediately see that the leading order term in the error is $\propto h^2\nabla^2 f(x,y)=h^4\nabla^4 u(x,y)$ so it will indeed vanish for Laplace's equation, and we will have a stencil with error $\mathrm{O}(h^4)$. But the crucial thing to note is that $f(x,y)$ is a known function, so we can actually calculate the h^2 term in **(2.10)** explicitly. Hence, it is possible to modify our stencil in such a way as to cancel this error term:

```
> stencil[2] := stencil[2]-convert(Error[poisson],polynom);
```

$$stencil_2 := \frac{1}{6}\frac{u(x-h,y-h)}{h^2} + \frac{2}{3}\frac{u(x,y-h)}{h^2} + \frac{1}{6}\frac{u(x+h,y-h)}{h^2} + \frac{2}{3}\frac{u(x-h,y)}{h^2} - \frac{10}{3}\frac{u(x,y)}{h^2} + \frac{2}{3}\frac{u(x+h,y)}{h^2} + \frac{1}{6}\frac{u(x-h,y+h)}{h^2} + \frac{2}{3}\frac{u(x,y+h)}{h^2} + \frac{1}{6}\frac{u(x+h,y+h)}{h^2} - f(x,y) - \frac{1}{12}h^2\left(\frac{\partial^2}{\partial y^2}f(x,y)\right) - \frac{1}{12}h^2\left(\frac{\partial^2}{\partial x^2}f(x,y)\right) \quad (2.11$$

If the source is known analytically, we could in principle calculate the derivatives directly. But the source might not be known analytically (i.e., we could only have numeric knowledge), or we may be coding in an environment where we cannot take the derivative automatically (i.e., in C or FORTRAN). At any rate, we only need the derivatives to enough accuracy to negate the leading order term in **(2.10)**, so we can use the followed centered stencils:

```
> substencil[4] := GenerateStencil(diff(f(x,y),x,x),3);
  substencil[5] := GenerateStencil(diff(f(x,y),y,y),3);
```

This stencil is of order 2

$$substencil_4 := \frac{\partial^2}{\partial x^2}f(x,y) = \frac{f(x-h,y)}{h^2} - \frac{2f(x,y)}{h^2} + \frac{f(x+h,y)}{h^2}$$

This stencil is of order 2

$$substencil_5 := \frac{\partial^2}{\partial y^2}f(x,y) = \frac{f(x,y-h)}{h^2} - \frac{2f(x,y)}{h^2} + \frac{f(x,y+h)}{h^2} \quad (2.12$$

Subbing these in to **(2.11)** yield the final form of the nine-point stencil of the Poisson equation:

```
> stencil[2] := expand(subs(substencil[4],substencil[5],stencil[2]));
```

$$stencil_2 := \frac{1}{6}\frac{u(x-h,y-h)}{h^2} + \frac{2}{3}\frac{u(x,y-h)}{h^2} + \frac{1}{6}\frac{u(x+h,y-h)}{h^2} + \frac{2}{3}\frac{u(x-h,y)}{h^2} - \frac{10}{3}\frac{u(x,y)}{h^2} + \frac{2}{3}\frac{u(x+h,y)}{h^2} + \frac{1}{6}\frac{u(x-h,y+h)}{h^2} + \frac{2}{3}\frac{u(x,y+h)}{h^2} + \frac{1}{6}\frac{u(x+h,y+h)}{h^2} - \frac{2}{3}f(x,y) - \frac{1}{12}f(x,y-h) - \frac{1}{12}f(x,y+h) - \frac{1}{12}f(x-h,y) - \frac{1}{12}f(x+h,y) \quad (2.13$$

We confirm that the error in this stencil is of order h^4 for the *Poisson* equation (not just the Laplace equation):

```
> Error[poisson] := dsubs(convert(derivative,list),convert(convert(series(stencil[2],h,8),`+
  `),diff));
```

$$Error_{poisson} := -\frac{1}{240} h^4 \left(\frac{\partial^4}{\partial y^4} f(x,y)\right) + \frac{1}{90} h^4 \left(\frac{\partial^4}{\partial y^2 \partial x^2} f(x,y)\right) - \frac{1}{240} h^4 \left(\frac{\partial^4}{\partial x^4} f(x,y)\right) + O(h^6) \quad \textbf{(2.14)}$$

Note that even this error is only a functional of $f(x, y)$, not of the solution $u(x, y)$. Hence, we could have even cancelled this term in the error by adding more terms to **(2.13)** if an even more accurate stencil is desired. However, the finite difference representation of the higher order derivatives in **(2.14)** will require more than the nine points already in **(2.13)**, which is not desirable if we cannot take derivatives of f analytically.

The **PoissonSolve** procedure from above does not need to be modified to use the nine-point stencil (instead of the five-point) as long as we define a mapping **Stencil[2]** that yields the constraint among the $u_{i,j}$'s at a given lattice point:

```
> Subs := seq(seq(u(x+ii*h,y+jj*h)=u[i+ii,j+jj],ii=-1..1),jj=-1..1),seq(seq(f(x+ii*h,y+jj*h)
  =f[i+ii,j+jj],ii=-1..1),jj=-1..1):
  stencil[2] := subs(Subs,stencil[2]):
  Stencil[2] := unapply(stencil[2],h,i,j,u,f);
```

$$Stencil_2 := (h, i, j, u, f) \rightarrow \frac{1}{6} \frac{u_{i-1,j-1}}{h^2} + \frac{2}{3} \frac{u_{i,j-1}}{h^2} + \frac{1}{6} \frac{u_{i+1,j-1}}{h^2} + \frac{2}{3} \frac{u_{i-1,j}}{h^2} - \frac{10}{3} \frac{u_{i,j}}{h^2} + \frac{2}{3} \frac{u_{i+1,j}}{h^2} + \frac{1}{6} \frac{u_{i-1,j+1}}{h^2} + \frac{2}{3} \frac{u_{i,j+1}}{h^2} + \frac{1}{6} \frac{u_{i+1,j+1}}{h^2} - \frac{2}{3} f_{i,j} - \frac{1}{12} f_{i,j-1} - \frac{1}{12} f_{i,j+1} - \frac{1}{12} f_{i-1,j} - \frac{1}{12} f_{i+1,j} \quad \textbf{(2.15)}$$

Here is some example output:

```
> n := 2;
  r := 4;
  R := 7;
  L := 25;
  g := [x->0,x->0,x->0,x->0];
  f := (x,y) -> add((-1)^i*Heaviside(r^2-(x-R*cos(Pi*i/n))^2 - (y-R*sin(Pi*i/n))^2),i=0..2*
  n-1);
  p[1] := plot3d(f(x,y),x=-L..L,y=-L..L,grid=[100,100],shading=zhue,axes=boxed,style=
  patchcontour,labels=[x,y,`f(x,y)`]):
  p[2] := FivePoint(30,f,g,L,2):
  display(Array([p[1],p[2]]));
```

$$n := 2$$

$$r := 4$$

$$R := 7$$

$$L := 25$$

$$g := [x \to 0, x \to 0, x \to 0, x \to 0]$$

$$f := (x, y) \to add\left((-1)^i \text{Heaviside}\left(r^2 - \left(x - R\cos\left(\frac{\pi i}{n}\right)\right)^2 - \left(y - R\sin\left(\frac{\pi i}{n}\right)\right)^2\right), i = 0..2n - 1\right)$$